/**
 * Copyright (c) 2010, iDA MediaFoundry
 * All rights reserved.
 *
 * Redistribution and use in source and binary forms, with or without
 * modification, are permitted provided that the following conditions are met:
 *
 * - Redistributions of source code must retain the above copyright notice,
 *   this list of conditions and the following disclaimer.
 *
 * - Redistributions in binary form must reproduce the above copyright notice,
 *   this list of conditions and the following disclaimer in the documentation
 *   and/or other materials provided with the distribution.
 *
 * - Neither the name of iDA MediaFoundry nor the names of its contributors may
 *   be used to endorse or promote products derived from this software without
 *   specific prior written permission.
 *
 * THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS"
 * AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE
 * IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE
 * ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT HOLDER OR CONTRIBUTORS BE
 * LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR
 * CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF
 * SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS
 * INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN
 * CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE)
 * ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED OF THE
 * POSSIBILITY OF SUCH DAMAGE.
 */
package be.idamf.cortizone.ioc.container
{
	import flash.events.Event;
	import flash.events.EventDispatcher;
	import flash.utils.Dictionary;
	import flash.utils.describeType;
	
	import mx.core.IMXMLObject;

	[Event(name="complete", type="flash.events.Event")]
	
	public class BeanCan extends EventDispatcher implements IMXMLObject
	{
		public static var beans:Dictionary = new Dictionary();
		public static var bindings:Dictionary = new Dictionary();

		public static function getObjectByName(name:String):*
		{
			var bean:* = beans[name];
			
			if (bean == undefined)
				return undefined;

			if (bean is IBean)
				return bean.value;

			return bean;
		}

		public static function getObjectByType(type:String):*
		{
			var binding:BeanTypeReference = bindings[type];
			return (binding ? binding.value : undefined);
		}

		public static function addBean(name:String, bean:*):void
		{
			beans[name] = bean;
		}

		public static function disposeBean(name:String):void
		{
			var bean:* = beans[name];

			if (bean == undefined)
				return;
			
			if (bean is IBean)
				bean.dispose();
			else
				delete beans[name];
		}
		
		public static function addBinding(binding:BeanTypeReference):void
		{
			bindings[binding.type] = binding;
		}
		
		////////////////////////////////////////////////////////////////////////
		// Instance
		
		protected var asyncBeans:uint;
		
		////////////////////////////////////////////////////////////////////////
		// Constructor

		public function BeanCan()
		{
			super();
			asyncBeans = 0;
		}
		
		////////////////////////////////////////////////////////////////////////
		// IMXMLObject impl
		
		public function initialized(document:Object, id:String):void
		{
			var info:XML = describeType(this);
			var accessor:XML;
			var property:String;
			var bean:*;
			
			for each (accessor in info..accessor.(@access == "readwrite"))
			{
				property = accessor.@name;
				
				bean = this[property];
				
				if (bean is IBean && bean["name"])
					continue;

				addBean(property, bean);
				
				if (bean is IAsyncBean)
				{
					IAsyncBean(bean).addEventListener(Event.COMPLETE, bean_completeHandler);					
					++asyncBeans;
				}
			}
			
			if (asyncBeans == 0)
				dispatchEvent(new Event(Event.COMPLETE));
		}
		
		////////////////////////////////////////////////////////////////////////
		// Handlers
		
		protected function bean_completeHandler(event:Event):void
		{
			IAsyncBean(event.target).removeEventListener(Event.COMPLETE, bean_completeHandler);	
			
			if (--asyncBeans == 0)
				dispatchEvent(new Event(Event.COMPLETE));
		}
	}
}